
Επιχειρησιακή ΄Ερευνα

Ενότητα 2 - Λύσεις Ασκήσεων

Βελτιστοποίηση με ΗΥ - AMPL

΄Ασκηση 1

Σύνολα.

Το σύνολο των διαδρομών, Routes, με στοιχεία: ATH SKG,SKG IST,ATH IST . Χρησιμο-
ποιούμε τον δείκτη j για τις διαδρομές.
Το σύνολο των πτήσεων, Flights, με στοιχεία: FL ATH SKG,FL SKG IST . Χρησιμοποιο-
ύμε τον δείκτη k για τις πτήσεις.
Το σύνολο των τύπων των εισιτηρίων, Ticket types, με στοιχεία: Y,B,M . Χρησιμοποιούμε τον
δείκτη i για τους τύπους εισιτηρίων.

Παράμετροι.

capacity : η χωρητικότητα του αεροπλάνου,
price[i, j] : η τιμή του εισιτηρίου τύπου i για τη διαδρομή j, i ∈ Ticket types, j ∈ Routes,
max tickets[i, j]: μέγιστος αριθμός εισιτηρίων τύπου i για τη διαδρομή j, i ∈ Ticket types, j ∈
Routes,

route flight[k, j] =

{
1 , αν η πτήση k περιλαμβάνεται στη διαδρομή j
0 , αλλιώς

, k ∈ Flights, j ∈ Routes.

Μεταβλητές απόφασης.

number tickets[i, j]: αριθμός εισιτηρίων τύπου i για τη διαδρομή j, i ∈ Ticket types, j ∈ Routes.

Π.γ.π.

max
∑

i∈T icket types

∑
j∈Routes price[i, j] ∗ number tickets[i, j]

υπό
∑

i∈T icket types

∑
j∈Routes route flight[k, j] ∗ number tickets[i, j] ≤ capacity, k ∈ Flights,

0 ≤ number tickets[i, j] ≤ max tickets[i, j], i ∈ Ticket types, j ∈ Routes.

Δεδομένα.

capacity = 300,
τα price[i, j] δίνονται από τον πίνακα

i\j ATH SKG SKG IST ATH IST
Y 300 160 360
B 220 130 280
M 100 80 140

,

τα max tickets[i, j] δίνονται από τον πίνακα

i\j ATH SKG SKG IST ATH IST
Y 40 80 30
B 80 130 100
M 220 200 180

και τα route flight[k, j] δίνονται από τον πίνακα

k\j ATH SKG SKG IST ATH IST
FL ATH SKG 1 0 1
FL SKG IST 0 1 0

.

1



Το αρχείο

Airticket_Scheduling.mod

με το μοντέλο που θα δώσουμε στην AMPL είναι

##Airticket scheduling - Model

set Routes;

set Flights;

set Ticket_types;

param capacity;

param price{i in Ticket_types, j in Routes};

param max_tickets{i in Ticket_types, j in Routes};

param route_flight{k in Flights, j in Routes};

var number_tickets{i in Ticket_types, j in Routes};

maximize profit: sum{i in Ticket_types, j in Routes}

price[i,j]*number_tickets[i,j];

subject to Capacity_flight{k in Flights}: sum{i in Ticket_types, j in Routes}

route_flight[k,j]*number_tickets[i,j]<=capacity;

subject to Max_tickets{i in Ticket_types, j in Routes}:

0 <=number_tickets[i,j]<=max_tickets[i,j];

Το αρχείο

Airticket_Scheduling.dat

με τα δεδομένα που θα δώσουμε στην AMPL είναι

##Airticket scheduling - Data

set Routes:= ATH_SKG SKG_IST ATH_IST;

set Flights:= FL_ATH_SKG FL_SKG_IST;

set Ticket_types:= Y B M;

param capacity:=300;

param price: ATH_SKG SKG_IST ATH_IST:=

Y 300 160 360

B 220 130 280

M 100 80 140;

param max_tickets: ATH_SKG SKG_IST ATH_IST:=

Y 40 80 30

B 80 130 100

M 220 200 180;

param route_flight: ATH_SKG SKG_IST ATH_IST:=

FL_ATH_SKG 1 0 1

FL_SKG_IST 0 1 1;



Οι εντολές που θα δώσουμε στην AMPL είναι

reset;

option solver cplex;

model Airticket_Scheduling.mod;

data Airticket_Scheduling.dat;

solve;

display number_tickets;



΄Ασκηση 2

Σύνολα.

Το σύνολο των εκπομπόμενων μέσων, E, με στοιχεία: τηλεόραση (i = 1), ραδιόφωνο (i = 2).
Το σύνολο των έντυπων μέσων, T , με στοιχεία: περιοδικά (i = 3), εφημερίδες (i = 4).
Το σύνολο των όλων των μέσων, M , M = E ∪ T .

Παράμετροι.

ci : το κόστος μιας διαφήμισης στο μέσο i, i ∈ M ,
ei : η αποτελεσματικότητα μιας διαφήμισης στο μέσο i, i ∈ M ,
Mi : ο μέγιστος αριθμός επιτρεπόμενων διαφημίσεων στο μέσο i, i ∈ M ,
mi : ο ελάχιστος αριθμός επιτρεπόμενων διαφημίσεων στο μέσο i, i ∈ M ,
c : το ποσό που μπορεί να δαπανηθεί σε διαφημίσεις,
p : το μέγιστο ποσοστό διαφημίσεων σε έντυπα μέσα επί του αριθμού εκπομπόμενων διαφημίσεων.

Μεταβλητές απόφασης.

xi: αριθμός διαφημίσεων στο μέσο i, i ∈ M .

Το π.γ.π. είναι

max
∑

i∈M eixi

υπό
∑

i∈M cixi ≤ c,
mi ≤ xi ≤ Mi, i ∈ M,∑

i∈T xi ≤ p
∑

i∈E xi.

Δεδομένα.

Τα ci και ei δίνονται από τον πίνακα
i 1 2 3 4
ci 1000 100 600 400
ei 100 22 70 50

,

M3 = 12 και M1 = M2 = M = 4 = 1000 (στην πραγματικότητα δεν περιορίζουν),
m1 = m2 = 10 και m3 = m4 = 0,
c = 30000, p = 0.4.



Το αρχείο

Advertisement_Campaign.mod

με το μοντέλο που θα δώσουμε στην AMPL είναι

##Advertisement campaign - Model

set Print_Media;

set Air_Media;

set Media=Air_Media union Print_Media;

param cost{i in Media};

param efficiency{i in Media};

param max_advert{i in Media};

param min_advert{i in Media};

param max_cost;

param max_rate_print_air;

var number_advert{i in Media};

maximize Efficiency: sum{i in Media} efficiency[i]*number_advert[i];

subject to Max_cost:

sum{i in Media} cost[i]*number_advert[i]<=max_cost;

subject to Max_advert{i in Media}:

number_advert[i]<=max_advert[i];

subject to Max_rate_print_air:

sum{i in Print_Media} number_advert[i]<=

max_rate_print_air* (sum{i in Air_Media} number_advert[i]);

subject to Min_advert{i in Media}:

number_advert[i]>=min_advert[i];

Το αρχείο

Advertisement_Campaign.dat

με τα δεδομένα που θα δώσουμε στην AMPL είναι

##Advertisement campaign - Data

set Print_Media:= Magazine Journal;

set Air_Media:= TV Radio;

param: cost efficiency max_advert min_advert:=

TV 1000 100 1000 10

Radio 100 22 1000 10

Magazine 600 70 12 0

Journal 400 50 1000 0;

param max_cost:= 30000;

param max_rate_print_air:=0.40; 1 1;



Οι εντολές που θα δώσουμε στην AMPL είναι

reset;

option solver cplex;

model Advertisement_Campaign.mod;

data Advertisement_Campaign.dat;

solve;

display number_advert;


